A Derived BrowseEipManager


One area of the Edit in Place (EIP) abilities of Clarion that has always bothered me is the base EIP classes do not function correctly when the browse format is changed.   Especially when you consider that moving the fields of the current sort order to the left is a common practice.  The problem is that when the order changes the default EIP class is not aware of the change and in blissful ignorance attempts to edit the wrong field.  Since I was unable to think of a valid reason why EIP should not function, regardless of the position of the fields in the browse, I decided to fix the problem.  The result of that decision is a derived BrowseEipManager class and this short and very informal essay.  


I do sixty-five to seventy percent of my programming with paper and pencil.  During that process I write notes, a lot of them, the majority of what you will read here is taken directly from those notes.  For the most part I just typed the notes into a Word document and used the spell checker; very little editing was done.   The point of all that is this, if you are expecting the quality of writing found in books and magazines you are in the wrong place. If you have problems with the text or do not understand something feel free to send me a note.   If you do not like the grammar or the bad spelling offends you, do not read the text.   A couple more additional comments and I will get started.  The solution I used to solve this particular problem is mine and is based on my opinions.  As with all programming problems there are several other ways to accomplish the task some maybe better solutions others may not.   In some places I just slip across the top of the code and in others I go line by line.   I am going to assume you are familiar with adding EIP to a browse, using the column specific options, disabling columns and the other standard tasks handled by the templates.   In addition, I assume you have at least a working knowledge of the Clarion language and understand how the FieldPairsClass functions.  If you are not comfortable with any of those topics I would suggest you read the relevant parts of the LRM and the AH, specifically the areas on the GET and PUT statements and the FieldPairsClass.  Those topics are covered nicely in the documentation and I am not going to review then here.   Okay, enough nonsense lets get started. 

The following was taken from a browse procedure with EIP enabled using the default instance class names, the code is placed in the ThisWindow.Init method by the templates.  Notice the sequence of the method calls, they are in order 1 through 7.  Also notice there is not a call with a sequence number three and the forth call is missing the first parameter.   More on these two calls later.  

BrwEquip.AddEditControl(EditInPlace::Tru:Code,1)  BrwEquip.AddEditControl(EditInPlace::Tru:UnitNum,2)  

BrwEquip.AddEditControl(,4)  

BrwEquip.AddEditControl(EditInPlace::Tru:Year,5)  BrwEquip.AddEditControl(EditInPlace::Tru:Make,6)  BrwEquip.AddEditControl(EditInPlace::Tru:Model,7)

These controls are added to a property of the BrowseClass called EditList, which is a reference to a queue.   In addition, the BrowseEipManager contains a property called EQ that refers to this queue.   Later in the text when I get to the Kill methods of the BrowseClass and the BrowseEipManager I will discuss why this is needed, for now just remember that the two properties refer to the same queue.   Now back to the ThisWindow.Init method.  Before the method calls AddEditControl the templates add each of the fields from the browse to a BrowseClass property called Fields.  Fields is an instance of the FieldPairsClass used by the BrowseEipManager to edit the list box.   When using EIP a queue element is being edited not the field from the file.  Actually the right side of the FieldPairsClass which is a reference to the queue element is what is being edited, more to follow.   

BrwEquip.AddField(Tru:Code,BrwEquip.Q.Tru:Code)

BrwEquip.AddField(Tru:UnitNum,BrwEquip.Q.Tru:UnitNum)

BrwEquip.AddField(Tru:VIN,BrwEquip.Q.Tru:VIN)

BrwEquip.AddField(Tru:License,BrwEquip.Q.Tru:License)

BrwEquip.AddField(Tru:Year,BrwEquip.Q.Tru:Year)

BrwEquip.AddField(Tru:Make,BrwEquip.Q.Tru:Make)

BrwEquip.AddField(Tru:Model,BrwEquip.Q.Tru:Model)

BrwEquip.AddField(Opr:Name,BrwEquip.Q.Opr:Name)

BrwEquip.AddField(Equ:TrailerNum,BrwEquip.Q.Equ:TrailerNum)

BrwEquip.AddField(Tru:Description,BrwEquip.Q.Tru:Description)


There are ten calls to the AddField method and only six calls to the AddEditControls method.  The last three calls, Opr:Name, Equ:TrailerNum and Tru:Description are actually ‘Hot Fields’ and can not be edited using the standard EIP classes.   They are added to the Fields property and the BrowseClass will handle the details of updating and displaying but for EIP they are of little interest.   The file field is added to the left side of the Fields and the queue element is added to the right side (see the AH for information on the FieldPairsCLass).  However, notice the Tru:Vin and the Tru:License fields that are added, the third and forth fields.  The Tru:Vin field is added to the Fields property but a call to AddEditControl for this field not present, the field is not added to the EditList property, but this field may be edited using EIP.   The Tru:License field, the forth field, is added to the Fields property and a call is made to AddEditControl is also made but the first parameter is missing.    When the AddEditControl method is called with the first parameter omitted the BroweEipManager adds a queue element to the EditList but the Control field of the queue is set to NULL.   A NULL value in this field identifies a disabled column for EIP and the field can not be edited using EIP.   Shortly I will be discussing the InitControls and some other methods of the BrowseEipManager and exactly how all this is done will be covered.    


A brief review and then I will get to the interesting stuff.  Fields are added to a property of the BrowseClass called Fields in the sequence they are placed in the list box.  Then controls are added to the EditList property of the BrowseClass in the sequence they are added to the list box.    That folks is the problem, when the BrowseEipManager starts up and begins editing fields in a list box the class expects everything to be in the order that it was originally added.  Unfortunately if the list box format has been changed the fields and controls will be out of sync.   

So all that is needed is some technique of telling the BrowseEipManger the order has been changed and have the class edit the correct fields.  Sounds simple and it actually is simple.

Before the sequence problem can be corrected I need to review a couple of the BrowseEipManger methods, specifically the AddControl and InitControls, once these two methods are understood the rest is just a little code. 



The AddControl method is called from the BrowseClass AddEditControl method, the code from the BrowseClass follows, 

BrowseClass.AddEditControl PROCEDURE(EditClass EC,UNSIGNED Id,BYTE Free)

  CODE

    SELF.CheckEIP

    SELF.EIP.AddControl(EC,Id,Free). 

CheckEip is called and then AddControl method.   The CheckEip method is from the BrowseClass and is where the EditList and EQ references are assigned to the same queue (SELF.EIP.Eq &= SELF.EditList).  Other than that the CheckEip method sets some values for the BrowseEIPManager ArrowAction , TakeFocusLoss and a few others.   They do not have any affect on the sequence problem but if you are interested that is where the values are assigned.   The call to Self.Eip.AddControl calls the BrowseEipManager AddControl method. The parameters are EC, an optional parameter that defaults to NULL.  I have been referring to this as a control, it is actually a reference to a class but you can think of it as a control.   The Id is the sequence number of the control, an optional parameter that defaults to zero.  The parameter is always omitted when the column specific option is used.   


The AddControl method simply adds a queue element filled with the values from the parameters.  

EIPManager.AddControl PROCEDURE(EditClass EC,UNSIGNED Id,BYTE Free)

  CODE

    SELF.EQ.Control &= EC

    SELF.EQ.Field = Id

    SELF.EQ.FreeUp = Free

    ADD(SELF.EQ, +SELF.EQ.Field)

    ASSERT(~ERRORCODE()) 

The only thing that is of any particular interest in this method is the add statement.   The queue element is added in sorted order based on the value of the Id parameter.   Other than that the method just assigns the parameters to the queue elements.     

NOTE:  Because of the use of the private attribute the AddEditControl method must be called at least once.   This problem could be avoided by deriving the BrowseClass and adding two or three methods and property or two.  Since I have never used EIP without setting some options on at least one column I decided not to fool with deriving the browse.   If that particular fact bothers you then do not use the class.   

             
During a browse when the user decides to edit a record the browse class jumps through a couple of hoops and eventually gets around to calling the AskRecord method.  

BrowseClass.AskRecord PROCEDURE(BYTE Req)

  CODE

  SELF.CheckEIP

  RETURN SELF.EIP.Run(Req)

The CheckEip method, this is one of the private problems, is called again and then the Self.Eip.Run method is called.   Now the additional call to CheckEip may appear to be unnessary but it is needed if the browse is set up without any of the column specific options being used.   Granted the browse class could be made to track the calls to CheckEip but that would require an additional property and some ‘if else’ statements, probably more trouble than it would be worth.    The EIP Run method is called and the BrowseEipManager takes control of the screen and EIP is initialized.   


Run methods are normally associated with the WindowManager.   The BrowseEipManager is derived from the EipManager and that class is actually derived from the WindowManager.    The one thing you need to remember here is that once EIP has been called the browse class gives up control of the screen and the procedure is actually in a different accept loop.  I am not going to get into all the details involved with that but it is an important concept to remember.   The first thing the Run method does is make a call to the Init method.  The BrowseEipManager has an Init method and that is where things get interesting.   

EIPManager.Init PROCEDURE

  CODE

  IF SELF.Column = 0 THEN SELF.Column = 1.

  SELF.LastColumn = 0

  SELF.Repost = 0

  SELF.RepostField = 0

  ASSERT(~SELF.EQ &= NULL)

  SELF.EQ.Field = 1

  SELF.InitControls

  SELF.ResetColumn

  RETURN Level:Benign

For the most part this method sets some start up values and returns, the two lines that are important are the Self.Eq.Field = 1 and the Self.InitControls. 

EIPManager.InitControls PROCEDURE

  CODE

1.   ASSERT(~SELF.Fields &= NULL)

2.   LOOP WHILE SELF.ListControl{PROPLIST:Exists,SELF.EQ.Field}

3.      GET(SELF.EQ,SELF.EQ.Field)

4.      IF ERRORCODE()

5.         SELF.EQ.Control &= NEW EditEntryClass

6.         SELF.AddControl(SELF.EQ.Control,SELF.EQ.Field,1)

7.      END

8.      GET(SELF.Fields.List,SELF.EQ.Field)

9.      ASSERT(~ERRORCODE())

10.    IF ~SELF.EQ.Control &= NULL

11.       SELF.EQ.Control.Init(SELF.EQ.Field,SELF.ListControl,SELF.Fields.List.Right)

12.    ELSE

13.       SELF.ListControl{PROPLIST:textcolor,SELF.EQ.Field} = COLOR:GRAYTEXT

14.    END

15.    SELF.EQ.Field += 1

16. END

I added the numbers to the left of this method because shortly when I derive the class this method is going to be replaced in the derived class.   I also need to get into the details of how this method functions and the line numbers make easy reference points. 

Line 1.   Just throw an error message, if the Fields property is NULL.   If this property is empty EIP will not function because there is nothing available to edit.   Notice the Self.Fields inside of the BrowseEipManager.  Take a look at the BrowseClass CheckEip method the Fields property from the browse is assigned to the Fields property of the BrowseEipManager.  Again the use of more than one reference to the same item.   

Line 2.   Start a loop and loop through each field in the list box.   Starting at the first field of the list box.   Remember the Field variable was set to one in the Init method.  

Line 3.    Get the EQ queue element that is at location Self.Eq.Field.   See the LRM if you are not comfortable with the GET statement and using the various parameters.   

Line 4. Test for errorcode.  If the GET statement posts an error in this situation that means there has not been a call to the AddControl method for this field location.   Remember back at the start when the Tru:Vin field was added to the browse but not added to the EditList, the third field.   When the InitControls method does not find a control it will create a default EditEntryClass control for use by the column.   

Line 5 and 6.   These two lines use the new statement to create a new control and call the AddControl method to add the newly created control to the queue.   Notice in line 6 the optional Free parameter is used here and is set to one, previously the default value of zero was used.   Later in the Kill method this will become critical.   Backtracking a little.  In the AddControl method the queue elements were added in a sorted order, based on the value of the Id parameter.

The same happens here what you need to know is the add statement used with queues and the optional parameter.  When duplicate values are found the new element is added after an existing one.   This allows controls to be created at run time and added to the EQ list.   Again, if that does not make sense see the LRM and the ADD statement.  

Line 8.  GET the Self.Fields.List element at location Self.Eq.Field.   Remember that the BrowseEipManager and the BrowseClass Fields property refer to the same item.  Also remebr that these were added in the ThisWindow.Init method with calls to the BrowseClass AddField method and those calls were in sequence.    

Line 9.  Assert if the get fails.

Line 10.   Test the value of the Control field, if NULL then the EQ element was added by a call to AddControl with the first parameter missing.   This tells the BrowseEipManager to disable the column and that happens in the else clause, line 12.   The color of the column is set to gray and the manager will ignore this column. 

Line 11.   If the control is not NULL then Init the control with the parameters.   Notice the SELF.Fields.List.Right parameter this is the right side of the FieldPairsClass list element and contains a reference to the queue element for the current column.   This parameter is used as prop:use variable for the newly created control, that is why the editing actually happens to the queue element not the file field.   If you are interested then see the Init method for the EditClass.  I am not going any further into the method because it does not affect what I am attempting to accomplish and the text is getting long enough without anymore details.   

Line 15, increment the Self.Eq.Field and start the loop over. 

A brief recap and then I will derive the class.   There is nothing complex about what this method accomplishes.  All the method does is test for a valid field in the list box, if the field is there, read two queue elements and use the values from those two queue elements to create a field that can be edited.   That is it.   The problem is the method wants everything in sequence and we want to change that sequence.   And now finally the fix.


Actually there are two problems the BrowseClass Fields property and the EIP manager EQ property.  Since the both are read in sequence all that is required is some way of reading the then in a different order.   The Fields problem is the simpler of the two to solve.  The Fields are added in the sequence that the list box is created and the list box uses a number to track the queue element for formatting.    That number is stored in the format string of the list box and can be read by using the proplist:fieldno property.   These numbers do not change when the order of the format changes.   Below is a sample format string from a list box.  The string is normally one long value but I shortened it broke it up here for easy reading. 

26L(2)|_F~CODE~@s2@38L(2)|_F~EQUIP ID~@s5@#1# 

84L(2)|_~VIN~@s20@#2#

44L(2)|_~LICENSE~@s1@#3#

30C|_~YEAR~@n4b@#4#

The numbers at the end of each line surrounded by the # are the ones needed.   These numbers do not change and the are the values returned by the proplist:fieldno property.   That will solve the Fields problem, all that is needed is to read the fieldno from the list box and use that value to GET the Fields.List.  The Fields.List will then contain a reference to the queue element that matches the column for EIP.  


The EQ problem is a little more complicated, they queue elements are added in sequence and that order needs to be changed.  So how?   I decided to use the example from the ABC libraries and use another queue reference variable and refer to the same locations in memory. 

Here is the header of the class, with the some of the attributes removed. 

SwapEIPManager   CLASS(BrowseEIPManager),TYPE… 

SwapEq                       &EditQueue

InitControls                  PROCEDURE,VIRTUAL

InitSwapQueue            procedure

KillSwapQueue           procedure

SwapControls              procedure,private

                                END

The SwapEq property is another reference to a EditQueue.   The four methods handle the details of initialization and altering the order of the queue elements.   Alright lets look at each of these four methods and see what happens and why.  

SwapEipManager.InitSwapQueue procedure

Count   long,auto

  code

  ! create new queue

1.   Self.SwapEq &= new(EditQueue)

2.   Self.Eq.Field = 1

3.   loop while Self.ListControl{proplist:Exists,Self.Eq.Field}

4.     get(Self.Eq, Self.Eq.Field)

5.     if (errorcode())

6.       Self.Eq.Control &= NULL

7.        Self.AddControl(Self.Eq.Control, Self.Eq.Field, 0)

8.        Self.SwapEq.Control &= new(EditEntryClass)

9.        Self.SwapEq.Field = Self.Eq.Field

10.      Self.SwapEq.FreeUp = true

11.     add(Self.SwapEq,  +Self.SwapEq.Field)

12.   else

13.       Self.SwapEq.Field = Self.Eq.Field

14.       Self.SwapEq.FreeUp = false

15.       Self.SwapEq.Control &= Self.Eq.Control

16.      add(Self.SwapEq, +Self.SwapEq.Field)

17.   end

18.  Self.Eq.Field += 1

19.  end

20.  return

Again line numbers are added for reference.    

The template that goes along with this class will call the InitSwapQueue from the BrowseClass SetAlerts method.   Since this is part of a derived BrowseEipManager the property can not be referenced through the EIP property of the BrowseClass, BrwX.EIP.SwapEq.  The compiler will complain and nothing will ever be solved.   In these situations you can derive another class, the browse, and change the data type of the EIP property.  I decided not to do that in this case because I only need to access the SwapEipManager in two places, once at start up and once at shut down.  The browse procedure has an instance of the EIP manager and that instance maybe used to access the class.   Using a derived BrowseClass in this case is more trouble than it is worth.   There are some other things that could be done, involving abstract base classes, but they are considerably more complex than just embedding two lines of code.   Okay back to the InitSwapQueue method.

Line 1.   Create a new queue to hold the information.  

Line 2.   Prime the loop control variable.

Line 3.   Loop through the list box for each field. 

Line 4.  GET the EQ element for the current column.   Remember that EQ is created and filled early in the ThisWindow.Init method and that the BrowseClass SetAlerts method is called very late in the method. 

Line 5.  Test for errorcode.  If there is an error then the field was not added to the EQ property and would normally be created in the InitControls method.  In this class I am going to create the control here.  

Line 6.  Set the control element to NULL, just give it something to test.   

Line 7.  Call AddControl to add the element to the EQ list.  Note the optional Free parameter is set to 0.  I could have left it off and just used the default value but I added it to make it clear in the code what was going on.    This new queue element will be added in the correct sequence just as if it had been added from the ThisWindow.Init method.   

Line 8 – 11.  

Create a new EditEntryClass, set the Field number and the Free value and ADD the element to the SwapEq list.    

Lines 13 – 16.   The GET statement did not post an error so this field has already been added to the EQ list.  Make a copy of the fields and ADD that to the SwapEq field.  Take a look at this code before continuing and get comfortable with it.  The SwapEq is a different queue than the EQ.  Note the values of the Field elements are the same in both queues.  The Free element is set to 1 telling the Kill method later on to dispose of this Control.  The interesting part is the Control property, if no error was posted I am not creating another class instance but the Control element for the SwapEq and the EQ lists refer to the same instance.    

Line 18.  Increment the loop control variable and start over.  

Line 20.  Return.  

What you need to take out of this is that after a call to InitSwapQueue there are two queues in memory.   Part of these two queues hold different values and other parts hold the same value.   And each is in sequence with the fields of the list box.   What good does all that do?   The second line of the derived InitControls method is a call to a new method, SwapControls.     The SwapControls method does exactly what the name implies, it swaps the order of the controls around so they will match the current column position of the list box.    Now lets look at the InitControls method for the derived class an see what happens to the EIP. 

SwapEIPManager.InitControls PROCEDURE !VIRTUAL

FormatPos     long,auto

  code

  assert(~SELF.Fields &= NULL)

  Self.SwapControls()   ! set Control to correct location

  Self.Eq.Field = 1     ! start at one

  loop while Self.ListControl{proplist:Exists,Self.Eq.Field}

    get(Self.Eq, Self.Eq.Field)

    FormatPos = Self.ListControl{proplist:fieldno, Self.Eq.Field}

    get(Self.Fields.List, FormatPos)

    if  ~Self.Eq.Control &= NULL

        Self.Eq.Control.Init(Self.EQ.Field, Self.ListControl, Self.Fields.List.Right)

    else 

      Self.ListControl{PROPLIST:textcolor, Self.EQ.Field} = COLOR:GRAYTEXT

    end 

    Self.Eq.Field += 1

  end

  return

I am not going to go over this method in detail, it does pretty much what the parent method does with a couple of exceptions.   

The first thing done is a test for Fields and then a call to SwapControls.   The loop control variable is primed again otherwise we would be starting at the end of the list box.  

Then Get the EQ list for the value.  The EQ list has already been reordered so the control refers to a valid class type.   Now use the fieldno property again to find out what Fields element is needed.  

GET that Fields.List element.   The EQ list and the Right side of the Fields property now refer to the correct edit class type and queue element for the current column of the list box.   The rest of the method is exactly like the parent method.  Test for a NULL value and initialize the control or disable the column.   

SwapEipManager.SwapControls procedure   ! private

FormatPos     long,auto

  code

1.   loop while Self.ListControl{proplist:exists, Self.Eq.Field}

2.     get(Self.Eq, Self.Eq.Field)

3.     FormatPos = Self.ListControl{proplist:fieldno, Self.Eq.Field}

4.     get(Self.SwapEq, FormatPos)

5.     Self.Eq.Control &= Self.SwapEq.Control

6.     put(Self.Eq)

7.    Self.Eq.Field += 1

8.  end

9. return

Line 1.   Loop through the fields in the list box.

Line 2.   GET the EQ element for the current Self.Eq.Field value.  This value is primed before the method call.

Line 3.  Set the local variable FormatPos equal to the field number form the list box property.  

Line 4.  GET the element from the SwapEq based on the current field number.   Remember SwapEq was created in the original sequence but the list box format order may change, but the field number of the list box format does not change.  What is happening is the method is reading the EQ list in sequence and jumping around the SwapEq list.  

Line 5.   Assign the EQ Control the reference for the SwapEq.   

Line 6.  Store the EQ list back in the original location but with the reference now pointing to a control from the SwapEq.

Line 8 and 9 Repeat the loop and return when done.  

The end result is EQ is still in its original order but the Controls now point to the correct classes 

for the new column location. 


The other method of the derived class, Kill, does not do anything particular interesting it simply steps through the SwapEq list and disposes any Controls created with the NEW statement and then disposes the SwapEq reference.     


In closing I hope you enjoy reading this text and with a little luck learned something  about the ABC classes.   I know I learned a great deal about the various classes used in the EIP  process.    I apologize for any spelling or grammar errors that caused problems, I am not a writer by trade and to be honest I am not going to spend the additional time and energy to edit my notes, just not going to do it folks.


Two final comments and I will leave you alone.    Hopefully after reading this text you will see that the ABC classes are not complex and they do not contain any black magic, just plain old Clarion code.  A little study and reading is all that is needed.   And finally if you do not gain anything else from this text, remember the following.   

The EQ property that stores the EIP information is a queue, since it is a queue it is created and destroyed at run time, a queue maybe manipulated at run time.  Filters, Range Limits, Sort Orders and the majority of the information used by the ABC classes are stored in a queue somewhere.   Those pieces of information are created and destroyed at run time also, but more importantly they can be manipulated at runtime.   The next time the default ABC behavior is not what you want, take look under the hood, chances are pretty good you will find a queue and that queue belongs to you the programmer.

